
Purpose

We build for the euro area the dataset we built for the US in Part A: a policy rate, inflation and a
measure of slack, all quarterly. The result is taylor_ea.dta, which we will use in Lab 2 to compare
the ECB with the Fed. The steps are the same as in Part A; only the source changes.

1. The ECB Data Portal

1. Go to https://data.ecb.europa.eu. This is the ECB’s statistical database; it is free and
needs no account.

2. Type HICP in the search box and open HICP – Overall index, Euro area (changing composi-
tion), Annual rate of change. Look at the chart, the data table and the information about
the series.

3. Every series has a key, shown on its page. The key describes the series:

HICP M U2 N 000000 4D0 ANR

dataset monthly euro area not seasonally all items source: annual rate
(changing composition) adjusted Eurostat of change

Replacing one part of the key gives a related series: XEF000 instead of 000000 gives core inflation
(excluding energy, food, alcohol and tobacco); INX instead of ANR gives the index level.

Q1. Why do we use the euro area in its changing composition (U2) rather than a fixed group of
countries?

Answer. The ECB sets one policy rate for the countries that are members at each date: 11 in 1999,
21 since Bulgaria joined in 2026. The changing composition matches the economy the ECB was
actually responsible for at each point in time.

2. The series we need

Variable Series key Frequency, period

Overnight rate
(EONIA)

FM.Q.U2.EUR.4F.MM.EONIA.HSTA quarterly, 1999–2021*put this in
Google search as ECB search
does not find it

Overnight rate
(€STR)

FM.Q.U2.EUR.4F.MM.UONSTR.HSTA quarterly, from 2019Q4

HICP, headline
(index)

HICP.M.U2.N.000000.4D0.INX monthly

HICP, core (index) HICP.M.U2.N.XEF000.4D0.INX monthly
Real GDP (volume
index)

MNA.Q.Y.I9.W2.S1.S1.B.B1GQ._Z._Z._Z.
IX.LR.N

quarterly, seasonally adjusted

Output gap AMECO series AVGDGP, euro area, at
https://db.nomics.world/AMECO/AVGDGP

annual

1

https://data.ecb.europa.eu
https://db.nomics.world/AMECO/AVGDGP


Search for each key on the portal and look at the chart before going to Stata.

Which interest rate? The ECB has three official rates. The main refinancing operations rate
(MRO) is the rate at which banks borrow from the ECB for one week, against collateral, in its
regular weekly operations. The deposit facility rate (DFR) is the rate banks earn on money they
deposit overnight at the ECB. The marginal lending facility rate is the rate at which banks can
borrow overnight from the ECB, against collateral, whenever they need to. Together they form a
corridor: no bank lends to another below the DFR, since it can earn the DFR at the ECB, and
no bank borrows above the marginal lending rate, since it can borrow at that rate from the ECB.
The DFR is the floor, the marginal lending rate is the ceiling, and the MRO lies in between.

Q2. When was the euro-area overnight rate below zero? What does this imply for estimating a
Taylor rule?

Answer. From about 2015 to mid-2022: the DFR was cut below zero in June 2014 and reached
−0.50% in 2019, and it was raised back to zero only in July 2022. During these years the ECB also
used asset purchases and forward guidance, so the overnight rate does not capture the full policy
stance. As with the US zero lower bound, we will have to decide in Lab 2 whether to exclude these
years.

Inflation. The ECB targets headline inflation in the Harmonised Index of Consumer Prices
(HICP), year on year. The HICP is a consumer price index: it measures the prices of goods
and services bought by households, not producer prices or the prices of everything produced in the
economy (the GDP deflator).

• How it is calculated. Each national statistical office collects prices and computes its coun-
try’s HICP with the same, harmonized methods, so the indices are comparable across coun-
tries. Within each country, items are weighted by their share in household consumption
expenditure; the weights are updated every year.

• How countries are weighted. Eurostat combines the national HICPs into the euro-area in-
dex. Each country’s weight is its share in total euro-area household consumption expenditure,
also updated every year: Germany has about a quarter of the index, Greece about 2%.

The HICP is a monthly index, so we average it to quarters in Stata, as FRED did for us in Part A.

Q3. What is the most important difference between the HICP and the US CPI?

Answer. The HICP excludes the costs of owner-occupied housing, while the US CPI includes them
through owners’ equivalent rent, about a quarter of the index and its largest single item. When
housing costs rise faster than other prices, as in the US after 2021, US CPI inflation is pushed up
relative to the HICP. Other differences are smaller: the HICP covers spending in the country by
both residents and tourists, and its weights are updated every year.

Output gap. There is no euro-area equivalent of CBO’s quarterly potential output. The European
Commission estimates potential output with a production function, as CBO does, but publishes it
only once or twice a year, at annual frequency, in its AMECO database. For each country, potential
output combines the trend of labor input (working-age population, trend participation, the non-
accelerating wage rate of unemployment and trend hours), the capital stock, and the trend of total
factor productivity, using a Cobb–Douglas production function; the output gap is the percentage
deviation of actual from potential GDP, and the euro-area gap is the gap between total euro-area
GDP and total euro-area potential GDP, which equals a weighted average of the country gaps, each
country weighted by its share of euro-area potential output (unlike the HICP, where countries count
in proportion to their household consumption). We assign each year’s gap to all four quarters of
that year. We download it from DBnomics, a free site that republishes AMECO and other official
databases with a simple CSV download. The series also contains the Commission’s forecasts for the
coming years; they drop out when we merge with GDP.

Q4. What is lost by using an annual output gap in a quarterly regression?

Answer. Within a year the gap is constant, so it cannot capture movements from quarter to
quarter, for example the collapse in 2020Q2 and the rebound in 2020Q3. The coefficient on the gap
is estimated from year-to-year variation only.

Download. For each ECB series in the table: paste its key into the search box of the portal, open
the series, look at the chart, then click Download and choose CSV. Save the file in your lab1 folder
with the name below. Open one file in a text editor or Excel before going to Stata: it has a date
column, a period column and a value column. For the output gap, go to https://db.nomics.
world/AMECO/AVGDGP, open the series for the euro area, look at the chart and download it as CSV;
this file has two columns, the year and the value.

Series Save as

EONIA ecb_eonia.csv
€STR ecb_estr.csv
HICP, headline ecb_hicp.csv
HICP, core ecb_hicp_core.csv
Real GDP ecb_gdp.csv
Output gap (from DBnomics) ameco_gap.csv

3. Into Stata

Open a new do-file, lab1_ea.do, in the same folder as Part A. Each ECB file has three columns:
the date, the period and the value. The command rename (*) (date period name) gives them
short names in one step; if describe shows a different number of columns, rename them one by
one.

3.1 The policy rate

clear all
cd "C:/Users/yourname/Desktop/lab1" // change to your folder

* EONIA, quarterly
import delimited using "ecb_eonia.csv", clear case(lower)
describe
rename (*) (date period eonia)
gen qdate = qofd(date(date, "YMD"))
format qdate %tq
keep qdate eonia
save "eonia.dta", replace

* euro short-term rate, quarterly
import delimited using "ecb_estr.csv", clear case(lower)
describe
rename (*) (date period estr)
gen qdate = qofd(date(date, "YMD"))
format qdate %tq
keep qdate estr
save "estr.dta", replace

3.2 Inflation: from monthly to quarterly

* headline HICP, monthly index
import delimited using "ecb_hicp.csv", clear case(lower)
describe
rename (*) (date period hicp)
gen qdate = qofd(date(date, "YMD"))
collapse (mean) hicp, by(qdate)

format qdate %tq
save "hicp.dta", replace

* core HICP, monthly index
import delimited using "ecb_hicp_core.csv", clear case(lower)
describe
rename (*) (date period hicp_core)
gen qdate = qofd(date(date, "YMD"))
collapse (mean) hicp_core, by(qdate)
format qdate %tq
save "hicp_core.dta", replace

What it does. Each monthly date is converted to the quarter it belongs to. collapse (mean)
then replaces the monthly rows by one row per quarter, the average of its three months: the same
Average aggregation we chose in FRED.

3.3 GDP and the output gap

import delimited using "ecb_gdp.csv", clear case(lower)
describe
rename (*) (date period gdp)
gen qdate = qofd(date(date, "YMD"))
format qdate %tq
keep qdate gdp
label var gdp "Real GDP, volume index"
save "gdp_ea.dta", replace

* annual output gap from AMECO, via DBnomics (columns: year, value)
import delimited using "ameco_gap.csv", clear case(lower)
describe
rename (*) (year gap)
save "ameco_gap.dta", replace

3.4 Merge, build the variables, save

use "eonia.dta", clear
merge 1:1 qdate using "estr.dta", nogenerate
merge 1:1 qdate using "hicp.dta", nogenerate
merge 1:1 qdate using "hicp_core.dta", nogenerate
merge 1:1 qdate using "gdp_ea.dta", nogenerate

gen year = year(dofq(qdate))
merge m:1 year using "ameco_gap.dta", keep(master match) nogenerate

tsset qdate

* policy rate: EONIA until 2019q3, then euro short-term rate + 0.085
gen i = eonia if qdate <= tq(2019q3)
replace i = estr + 0.085 if qdate >= tq(2019q4)
label var i "Euro-area overnight rate (%)"

* inflation, four-quarter rate
gen pi_hicp = 100*(hicp/L4.hicp - 1)
gen pi_hicp_core = 100*(hicp_core/L4.hicp_core - 1)
label var pi_hicp "HICP inflation, y/y (%)"
label var pi_hicp_core "Core HICP inflation, y/y (%)"

gen pigap = pi_hicp - 2
label var pigap "HICP inflation minus 2% target (pp)"
label var gap "Output gap, AMECO (%)"

4



gen i_lag = L.i

tsline i pi_hicp if qdate >= tq(1999q1), yline(0 2) name(ea, replace)

keep if qdate >= tq(1999q1) & !missing(gdp)
order qdate i i_lag pi_hicp pi_hicp_core pigap gap
save "taylor_ea.dta", replace

Q5. Plot i and pi_hicp. When did the ECB start raising rates in 2022, and how far had inflation
risen by then? Compare with the Fed.

Answer. The ECB first raised rates in July 2022, when euro-area HICP inflation was already about
9%; it peaked at 10.6% in October 2022. The Fed started in March 2022, about four months earlier,
with inflation also well above target. Both reacted late; the ECB later than the Fed.

Q6. The ECB’s target was “below, but close to, 2%” until July 2021 and a symmetric 2% since
then. Does it matter which we use for pigap?

Answer. For the estimated responses, no: in a regression with a constant, the target is absorbed
by the intercept. It matters only if we want to test whether the ECB’s behaviour changed in 2021,
which we will do in Lab 3.

